


Student Problem Corner

Editor's Note: The problem presented here is suitable for students studying pre-calculus. It explores the concepts of exponential function and quadratic formula. The problem is reprinted with permission from Resource Problems to Enhance the Teaching of Mathematics, University of Delaware, Newark, Delaware. This collection of senior high school mathematical problems is available for \$4 U.S. from Willard E. Baxter, Department of Mathematical Sciences, 501 Ewing Hall, Newark, Delaware 19716. Phone: (302)451-2653.

Problem

A wire hanging symmetrically across a road is anchored at the edges of the road on standards. The standards are 16 feet high. The road is 60 feet wide, and the height of the wire at the centre of the road is 10 feet above the road. Furthermore, after establishing the coordinate system shown, it is observed that the graph of the wire is

$$y = \frac{B^x + B^{-x}}{2} + 9$$



1. Find B as an algebraic expression.
2. What special relationship do the two values of B have?
3. Give the first five decimals in the decimal approximation of B.
4. Can a truck 10 feet wide and 14 feet high pass under the wire?

Solution

FORMULA

$$x = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}, \text{ where } ax^2 + bx + c = 0$$

STRATEGY

Since (30, 16) is a point on the curve, substitute the coordinates into the formula. Express the resulting equation in quadratic form and solve with the quadratic formula.

Question 1

ARGUMENT

$$y = \frac{B^x + B^{-x}}{2} + 9$$

$$16 = \frac{B^{30} + B^{-30}}{2} + 9$$

$$32 = B^{30} + B^{-30} + 18$$

$$B^{30} + B^{-30} = 14$$

$$B^{60} - 14B^{30} + 1 = 0$$

$$B^{30} = \frac{14 \pm \sqrt{14^2 - 4(1)(1)}}{2}$$

$$B^{30} = \frac{14 + \sqrt{192}}{2}$$

$$B = \sqrt[30]{\frac{14 + 192}{2}}$$

ANSWER

$$B_1 = \sqrt[30]{\frac{14 + 192}{2}} \quad \text{and} \quad B_2 = \sqrt[30]{\frac{14 - 192}{2}}$$

Question 2

Since $B^x + B^{-x} = (1/B)^x + (1/B)^{-x}$, we see that $B_2 = 1/B_1$.

JUSTIFICATION

$$B_2 = \frac{\sqrt[30]{(14 - 192)(14 + 192)}}{2(14 + 192)}$$

$$= \frac{\sqrt[30]{2}}{14 + 192}$$

$$= 1/B_1$$

ANSWER

$$B_2 = 1/B_1$$

Question 3


The appended program allows one to approximate the values of B using a bracket-and-halving algorithm. We first estimate

$$n = \frac{14 + \sqrt{192}}{2} = 13.92820$$

ANSWER

Using the algorithm, we find that $B_1 = 1.09176$ and $B_2 = 0.91601$

Question 4



ANSWER

We wish to approximate y when $x = 20$. Using the accompanying algorithm

$$y = \frac{B^{20} + B^{-20}}{2} + 9 = 11.98097$$

Since the truck is 14 feet high, it *cannot* pass under the wire. Therefore, the answer is no.

Program for the Algorithm

```
10 ' Intent: Approximate a root of a number.
20 ' Evaluate y, when x is 20, in the formula from the problem.
30 ' Variables:
40 ' A$ is the user's answer to input questions.
50 ' B is the maximum amount of error to be allowed in the calculation.
60 ' E is the index of the root.
70 ' I is the counter for calculating the power by repeated multiplications.
80 ' L is the lower bound for the root.
90 ' M is the midpoint of L and R.
100 ' N is the real number whose root is to be calculated.
110 ' P is the E'th power of T.
```

```

120 ' Q is the reciprocal of P.
130 ' R is the upper bound for the root.
140 ' T is the current approximation of the root that is being tested.
150 '
200 ' BEGIN
210 CLS
220 GOSUB 500 ' Enter values and test boundary for root
230 L = 1
240 IF R - L < B GOTO 290
250 GOSUB 700 ' Calculate an approximation for the root
260 GOSUB 800 ' Calculate the power of the approximation
270 IF P > N THEN R = M ELSE L = M
280 GOTO 240
290 ' ENDLOOP
300 PRINT
310 PRINT "The root is ";R
320 PRINT
330 PRINT "Do you want to evaluate the formula?"
340 A$ = " "
350 INPUT "Type Y for yes, N for no";A$
360 IF A$ = "Y" OR A$ = "y" THEN ELSE 400
370 GOSUB 900 ' Calculate y value in formula, given x as 20
380 PRINT
390 PRINT "When x = 20 in the formula, y = ";Y;" feet."
400 ' ELSE
410 ' ENDIF
420 END
430 '
500 ' SUBROUTINE-Enter values and test boundary for root
510 INPUT "What is the real number N whose root you want to find? N = ";N
520 INPUT "What is the index of the root ";E
530 INPUT "What is the maximum error, you want to allow in the answer ";B
540 A$ = " "
550 IF A$ = "Y" OR A$ = "y" GOTO 660
560 INPUT "Name a value that is an upper (or lower) bound for N";R
570 PRINT "The computer will now test the value you just entered."
580 PRINT
590 T = R
600 GOSUB 800
610 PRINT P
620 PRINT
630 PRINT "Is the value printed above larger (or smaller) than N?"
640 INPUT "Type Y for yes, N for no. ";A$
650 GOTO 550
660 ' ENDLOOP
670 RETURN
700 ' SUBROUTINE - Calculate an approximation of the root
710 M = (L + R)/2
720 T = M
730 RETURN
800 ' SUBROUTINE - Calculate the power of the approximation
810 P = 1

```

```

820   FOR I = 1 TO E
830     P = P*T
840   NEXT I
850   RETURN
900   'SUBROUTINE-Calculate y value in formula, given x as 20
910     T = R
920     E = 20
930     GOSUB 800
940     Q = 1/P
950     Y = (P + Q)/2 + 9
960   RETURN

```

This program was run on an IBM-PC.

RUN

What is the real number N whose root you want to find? N = ? 13.92820

What is the index of the root? 30

What is the maximum error you want to allow in the answer? .000001

Name a value that is an upper (or lower) bound for N? 1.5

The computer will now test the value you just entered.

191751.1

Is the value printed above larger (or smaller) than N?

Type Y for yes, N for no.? y

The root is 1.091767

Do you want to evaluate the formula?

Type Y for yes, N for no? y

When $x = 20$ in the formula, $y = 11.98086$ feet

